

11 Chapter: Fundamental Theorem of Finite Abelian Groups

Recall from Chapter 8 of this manual, the command to form the direct product of two or more groups in GAP is `DirectProduct`. For example the below output creates a group G isomorphic to $\mathbf{Z}_6 \oplus \mathbf{Z}_6$:

```
gap> Z6:= CyclicGroup(IsPermGroup,6);
Group([ (1,2,3,4,5,6) ])
gap> G:=DirectProduct(Z6,Z6);
Group([ (1,2,3,4,5,6), (7,8,9,10,11,12) ])
```

The group \mathbf{Z}_6 is a cyclic group of order 6. The group `Z6` in the above GAP commands is also a cyclic group of order 6 (namely, the subgroup of S_6 generated by the permutation $(1, 2, 3, 4, 5, 6)$). Thus \mathbf{Z}_6 is isomorphic to `Z6`.

Notice in the above output the elements in `Z6` that are in the second component are described as powers of the 6-cycle $(7, 8, 9, 10, 11, 12)$ instead of the 6-cycle $(1, 2, 3, 4, 5, 6)$. (The powers of $(7, 8, 9, 10, 11, 12)$ also form a cyclic group of order 6, so $\langle (7, 8, 9, 10, 11, 12) \rangle$ is isomorphic to \mathbf{Z}_6 as well.) GAP automatically changed the description of the elements in the second component of the direct summand from powers of $(1, 2, 3, 4, 5, 6)$ to powers of $(7, 8, 9, 10, 11, 12)$ since we are trying to represent an external direct product $H \oplus K$ as an internal direct product $H \times K$ and we need $H \cap K = \{e\}$.

By the Fundamental Theorem of Finite Abelian Groups every finite Abelian group is isomorphic to the direct product of cyclic groups of prime power order. We also know that a factor group G/H , where G is finite and Abelian, is also a finite Abelian group. Suppose $G = \mathbf{Z}_6 \oplus \mathbf{Z}_5 \oplus \mathbf{Z}_8$ and H is the subgroup of G generated by $(2, 1, 2)$. What finite Abelian group is G/H ? The following steps in GAP resolve this question.

```
gap> gap> Z5:= CyclicGroup(IsPermGroup,5);
Group([ (1,2,3,4,5) ])
gap> Z8:= CyclicGroup(IsPermGroup,8);
Group([ (1,2,3,4,5,6,7,8) ])
gap> G:= DirectProduct(Z6,Z5,Z8);
Group([ (1,2,3,4,5,6), (7,8,9,10,11), (12,13,14,15,16,17,18,19) ])
gap> H:= Subgroup(G, [(1,2,3,4,5,6)^2, (7,8,9,10,11),
>(12,13,14,15,16,17,18,19)^2]);
Group([ (1,3,5)(2,4,6), (7,8,9,10,11), (12,14,16,18)(13,15,17,19) ])
gap> F:= FactorGroup(G,H);
Group([ f1, f2 ])
gap> Size(F);
4
```

Note that \mathbf{Z}_5 is isomorphic to `Z5` and \mathbf{Z}_8 is isomorphic to `Z8`. Thus the `G` defined in the above GAP commands is isomorphic to $\mathbf{Z}_6 \oplus \mathbf{Z}_5 \oplus \mathbf{Z}_8$. Notice, in the above output, in the direct prod-

uct of \mathbf{Z}_6 , \mathbf{Z}_5 and \mathbf{Z}_8 , the elements of \mathbf{Z}_5 in the second component are written as powers of the permutation $(7, 8, 9, 10, 11)$. Similarly, the elements of \mathbf{Z}_8 in the third component are written as powers of the permutation $(12, 13, 14, 15, 16, 17, 18, 19)$. The element $(1, 2, 3, 4, 5, 6)^2$ generates a subgroup of order 3 in \mathbf{Z}_6 . Similarly $(12, 13, 14, 15, 16, 17, 18, 19)^2$ generates a subgroup of \mathbf{Z}_8 of order 4. Thus H is isomorphic to the subgroup of $\mathbf{Z}_6 \oplus \mathbf{Z}_5 \oplus \mathbf{Z}_8$ generated by the element $(2, 1, 2)$. The factor group is a finite Abelian group of order 4 so it must be isomorphic to either \mathbf{Z}_4 or $\mathbf{Z}_2 \oplus \mathbf{Z}_2$.

```
gap> Read("orderFrequency");
gap> orderFrequency(F);
[Order of element, Number of that order]=[ [ 1, 1 ], [ 2, 3 ] ]
```

Since the factor group has three elements of order 2 it must be isomorphic to $\mathbf{Z}_2 \oplus \mathbf{Z}_2$.

Let Z_n denote a cyclic group of order n . If m divides n , then Z_n contains a cyclic subgroup, Z_m , of order m .

Exercises

11.1 **By hand**, describe the cosets of $(Z_{48}/Z_6)/(Z_{12}/Z_6)$. Since this group is a finite Abelian group it is a direct product of cyclic groups of prime power order. Describe which one.

Attempting to work Exercise 11.1 in **GAP** is a little tricky. We would like to thank Andy Miller of Belmont University for providing the following explanation on how to use **GAP** to work Exercise 11.1. First enter the cyclic groups Z_{48} , Z_6 and Z_{12} into **GAP**:

```
gap> Z48:= CyclicGroup(IsPermGroup,48);;
gap> Z48.1;
(1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20,21,22,23,24,25,26,27,28,29,
30,31,32,33,34,35,36,37,38,39,40,41,42,43,44,45,46,47,48)
gap> Z12:= Subgroup(Z48,[Z48.1^4]);;
gap> Z6:=Subgroup(Z48,[Z48.1^8]);;
```

Then construct the factor groups Z_{48}/Z_6 and Z_{12}/Z_6 :

```
gap> g1:= FactorGroup(Z48,Z6);
Group([ f1, f2, f3 ])
gap> g2:= FactorGroup(Z12,Z6);
Group([ f1 ])
```

But when we then try to construct the factor group $(Z_{48}/Z_6)/(Z_{12}/Z_6)$ we get an error:

```
gap> FactorGroup(g1,g2);
Error, no method found! ...
```

The problem is the command **FactorGroup**(G,H) creates a group isomorphic (not equal) to G/H .

Thus `FactorGroup(Z12,Z6)` is not a subgroup of `FactorGroup(Z48,Z6)`. (`FactorGroup(Z12,Z6)` is only isomorphic to a subgroup of `FactorGroup(Z48,Z6)`). To express Z_{12}/Z_6 as a subgroup of Z_{48}/Z_6 we will use the fact that if N is a normal subgroup of a group G then the group G/N is the image of the homomorphism $\phi : G \rightarrow G/N$ defined by $\phi(g) = gN$. Thus, if H is a subgroup of G that contains N , the subgroup H/N of G/N is the image of H under ϕ . (That is, $H/N = \phi(H)$.) In the following GAP work we again have $g_1 = Z_{48}/Z_6$ and $g_2 = Z_{12}/Z_6$ but now g_1 and g_2 are constructed in such a way that g_2 is a subgroup of g_1 .

```
gap> phi:= NaturalHomomorphismByNormalSubgroup(Z48,Z6);;
gap> g1:= Image(phi,Z48);
Group([ f1, f2, f3 ])
gap> g2:= Image(phi,Z12);
Group([ f3 ])
gap> IsSubgroup(g1,g2);
true
```

Since g_2 is now a subgroup of g_1 we can use the `FactorGroup` command and determine the isomorphism class of g_1/g_2 .

```
gap> h:= FactorGroup(g1,g2);
Group([ f1, f2, <identity> of ... ])
gap> Size(h);
4
gap> Read("orderFrequency");
gap> orderFrequency(h);
[Order of element, Number of that order]=[ [ 1, 1 ], [ 2, 1 ], [ 4, 2 ] ]
```

Since g_1/g_2 is an Abelian group of order 4 with an element of order 4, g_1/g_2 is isomorphic to \mathbf{Z}_4 .

Exercises

11.2 Let $G = (Zr/Zs)/(Zt/Zs)$, where t divides r and s divides t . Make a conjecture about the isomorphism class of G . Prove your conjecture.

11.3 Let $G_1 = (Z_{24} \oplus Z_8 \oplus Z_{12})/(Z_2 \oplus Z_2 \oplus Z_2)$. Let $G_2 = (Z_8 \oplus Z_8 \oplus Z_6)/(Z_2 \oplus Z_2 \oplus Z_2)$. Use GAP to help you write G_1/G_2 as a direct product of cyclic groups of prime power order.

11.4 Repeat Exercise 11.3 for $G_1 = (Z_{40} \oplus Z_{10} \oplus Z_{24})/(Z_2 \oplus Z_5 \oplus Z_2)$ and $G_2 = (Z_{20} \oplus Z_{10} \oplus Z_6)/(Z_2 \oplus Z_5 \oplus Z_2)$.

11.5 Generalize your conjecture given in Exercise 11.2. That is, make a conjecture about the factor group $(G/H)/(K/H)$ when G, H and K are finite Abelian groups, H is a subgroup of K and H is a subgroup of G .